
Общие сведения

Язык  RCL (Reports Creation Language)  используется для создания форм отчетов и описания формата экспорта в программных модулях системы Maximus. Основное назначение языка – форматированное отображение результата вычислений.

Описание формы


Программный код RCL-формы хранится в текстовом файле или файле формата UNICODE-текст с расширением .FRM. 

Форма состоит из нескольких разделов (секций): управляющей секции Control, секции объявления переменных Declare и исполняемой части Begin-End. Кроме этого, в форму могут входить и другие секции. Название секции заключается в квадратные скобки, пробелы между скобками и названием не допускаются, прописные и строчные буквы система друг от друга не отличает.


Программный код RCL выполняется в режиме интерпритации.

Секция Control


Секция [Control] содержит разнообразные настройки, влияющие на просмотр и распечатку результата исполнения формы. Настройки представляют собой последовательность строк вида Параметр=Значение
Пример:

[Control]

FormName=Пример; Piemērs

Font=9

Заголовок формы под названием «Пример», размер шрифта по умолчанию внутри формы равен 9.

Секция Declare


Секция [Declare] содержит обявления переменных, используемых в форме. 


Пример:

[Declare]

HeaderFont="12.Arial Baltic Bold"
width=100

Эта секция не является обязательной в программном коде формы.
Секция Begin
Секция [Begin] состоит из исполняемых инструкций и форматных модулей, отвечающих за построение формы. Форматные модули представляют собой перечень параметров, разделенный запятыми и заключенный в фигурные скобки. Секция [Begin] обязательно должна заканчиваться выражением [End].

Пример:
[Begin]

{"Привет", width, HeaderFont, "B", "C"}

[End]

Комментарий


Программный код формы может содержать фрагменты, не участвующие в вычислениях. Они называются комментариями и обозначаются символом «//» в начале строки.

Пример:
[Begin]

// Это комментарий

[End]

Элементы языка RCL


В этом разделе описаны основные элементы  языка. 

Типы данных 

Символьный

Значения этого типа представляет собой последовательность из одного и более символов и обозначаются кавычками или апострофами.

Пример:

"Строка"
'Строка'
'  Это тоже "Строка"  '

Числовой


Тип данных идентифицирует величины, которые могут использоваться при выполнении математических операций. Числовые величины отображаются в программе цифрами от 0 до 9 с возможным указанием знака числа. Дробная часть числа отделяется точкой. 

Пример:

100

3.1415926

-200

// Шестнадцатеричная форма

0xFF

Логический 


Этот тип данных может иметь два значения .T.- истина и .F. - ложь.


Дата

Тип данных идентифицирует величины, представляющие календарные даты. Дата может быть определена как форматированная символьная строка, которая преобразовывается в величину типа «дата» специальной функцией CTOD().

Пример:

CTOD("01.01.2003")

// Пустая дата

CTOD("")

Nil

Nil-это пустое значение, не имеющее определенного типа. 


Блок кода

Тип данных блок кода определяет величину, содержащую исполняемый программный код. Используя этот тип данных, исполняемый программный код может быть присвоен переменной. В этом случае программный код выполняется всякий раз, когда переменная передается как параметр функции Eval()

Блок кода может содержать параметры, указанные между символами «|».

Пример:

[Declare]
Block:={ || "Привет" }

Block2:={ |name| "Привет, "+name }

[Begin]

{EVAL(Block)}

{EVAL(Block, "Maximus")}

[End]


Результат:

             


Привет




Привет, Maximus


Объект

Тип данных объект определяет величину, которая имеет собственные переменные и программный код. Определенный тип объекта называется классом. Переменные, входящие в объект, называются свойствами, сегменты программного кода носят название методы. Доступ к свойствам и методам производится указанием обекта и имени соответствующего элемента через двоеточие. 

Пример:

Obj:Name:= "Maximus"
Obj:Data()


Язык RCL не имеет возможности описания собственных классов обектов, но может пользоваться встроенными классами программного модуля, выполняющего форму.

Массивы

Типа данных массив является группой данных, которая может содержать одни и те же или различные типы данных. Используя массив, несколько величин могут присваиваться в одну переменную. Одна из величин массива называется элементом массива. Массив может быть инициализирован, как разделенный запятыми список, заключенный в фигурные скобки. Доступ к элементу массива производится при указании имени массива и номера элемента в квадратных скобках. Массивы могут быть многомерными.

Пример:

Array:={1,2,3,4}

Array[1]

Двумерный массив:

Array:={{1,2,3},{"one","two","three"}}

Константы            

Константами называются записанные в программном коде неизменные значения.

Пример:

     100.00


"Constant"
    
Nil


.T.   
Переменные
Переменная используется в программе, чтобы сохранить или извлечь какую-либо величину. В программном коде формы переменные представлены посредством идентификаторов.


Идентификатор переменной должен соответствовать следуюшим правилам:

· состоять только и символов латинского алфавита, цифр и символа “_”;

· имя переменной может начинаться с буквы или символа “_”;

Между прописными и строчными буквами идентификатора различий нет.
Для того, чтобы в программном коде формы можно было использовать переменную, она должна быть объявлена в секции [Declare]. Объявление может быть произведено простым указанием идентификатора. Также, при объявлении переменная может быть инициализирована. При инициализации переменной присваивается значение определенного типа.

Пример:

[Declare]

Text:= "Укажите номер счета: "
     Width:=10


x1


x2

  
x3:=Nil 

Переменной Text присвоена символьная строка, переменной Width присвоено числовое значение; переменные x1, x2 объявлены, но не инициализированы и имеют значение Nil; переменной x3 присвоено значение Nil.


Кроме объявляемых переменных в форме можно использовать ссылки на элементы встроенных объектов и поля баз данных, используемых программой, в рамках которой выполняется вычисление формы. 

Встроенные объекты


В некоторых случаях вызывающий программный модуль передает в форму объект некоторого класса. Доступ к элементам такого объекта производится при помощи оператора «::».

Пример:

::Date

Ссылка на переменную Date, элемент встроенного объекта.

Базы данных
Базы данных не являются непосредственным элементом языка, но представляют собой источник данных для RCL-формы. База данных состоит из записей, разделенных на именованные поля, содержащие данные символьного, числового, логического типа или типа «дата».

Обращение к значению поля базы даннных должно сопровождаться внутренним именем базы данных и оператором «->».

Пример:

Operat->Date

Ссылка на поле Date в базе данных Operat.

Операции

Операции являются существенным компонентом программного кода. Они выполняют определенные действия, используюя один или более операндов. Операнд – это константа, переменная или выражение, задействованное в операции. Последовательность символов, обозначающая операцию, называется оператором. Операторы могут быть использованы только с определенными, доступными для них типами данных.

В этом разделе и далее в описания будут использованы следующие обозначения

Тип данных


Обозначение

Символьный
C

Числовой

N

Логический 
L

Дата
D

Блок кода
B

Объект
O

Массивы
A

Неопределенный тип Nil 
U

Операции присвоения

Оператор
Типы данных
Название
Пример

=
Все
Присвоение
X=1

:=
Все
Присвоение
X:=1

+=
N, D, C
Встроенное сложение, конкатенация (слияние)
X+=1 аналог X:=X+1

-=
N, C
Встроенное вычитание, конкатенация
X-=1 аналог X:=X-1

*=
N
Встроенное умножение
X*=Y аналог X:=X*Y

/=
N
Встроенное деление
X/=Y аналог X:=X/Y

%=
N
Встроенное деление по модулю (остаток от деления двух чисел)
X%=Y аналог X:=X%Y

**= или ^=
N
Встроенное возведение в степень
X**=2 или X^=2

Математические операции

Оператор
Типы данных
Название
Пример

+
N
Префикс положительного значения числа
+1

+
N, D, C
Сложение, конкатенация
X+Y

-
N
Префикс отрицательного значения числа
-1

-
N, D, C
Вычитание, конкатенация без пробелов символьных строк
X-Y

*
N
Умножение
X*Y

/
N
Деление
X/Y

%
N
Деление по модулю 
X%Y

** или ^
N
Возведение в степень
X**2 или X^2

( и )
N
Изменение порядка вычисления выражения
X*(Y+Z)

Операции сравнения

Оператор
Типы данных
Название
Пример

==
Все
Эквивалентность
X==Y

=
Все
Эквивалентность *)
X=Y

<>  #   !=
N, C, D, L
Не эквивалентно
X!=Y  X<>Y  X#Y

>
N, C, D, L
Больше
X>Y

<
N, C, D, L
Меньше
X<Y

>=
N, C, D, L
Больше или равно
X>=Y

<=
N, C, D, L
Меньше или равно
X<=Y

$
C
Содержание одной строки в другой
X $ C

*) При сравнении символьных строк первая операнд сравнивается со вторым только по длине второго операнда. Например, результатом вычисления следующего выражения  будет «истина»: “ABC”=“AB”

Логические операции

Оператор
Типы данных
Название
Пример

!  или .NOT.
L
Логическое отрицание
!X .NOT. X

.AND.
L
Логическое «и»
X .and. Y

.OR.
L
Логическое «или»
X .or. Y

Специальные операции

Оператор
Типы данных
Название
Пример

@
Все
Передача параметра в функцию по ссылке
Test ( @p1 )

&
C
Выполнение макроподстановки
Macro:= «Test»

&(Macro) () аналог Test ()

:
O
Обращение к элементу объекта
Obj:Data

::
O
Обращение к элементу встроенного объекта
::Data

->
Все
Обращение к полю базы данных
Operat->Date

[ и ]
A, C
Обращение к элементу массива или символу строки
String[3]

Функции


Функцией называется определенный в исполняющем форму программном модуле фрагмент программного кода, производящий определенные действия и возвращающий результат. Каждая функция имеет название - идентификатор. Обращение к функции осуществляется путем указания в программном коде ее идентификатора в сопровождении круглых скобок. В круглых скобках могут быть указаны параметры, передаваемые функции. Если параметров несколько, то они разделяются запятыми. 
Пример:


[Declare]

     D1:=Date()


Debet:=TODeb ("2310", "001",  , "EUR", D1, D1+1)

Параметры функции могут быть обязательными и необязательным. Далее в описании  для обозначения необязательных параметров будут использоваться квадратные скобки. Если параметр или возвращаемая функцией величина могут приобретать только определенные значения, то эти значения будут перечисляться через вертикальную черту “|”.
Форматные модули


Форматным модулем называется элемент языка, формирующий графическое отображение результатов вычислений. Форматный модуль обрамляется фигурными скобками и  имеет следующую структуру:

 {[выражение],[ширина],[шрифт],[аттрибуты],[выравнивание]}
Выражение – значение числового, символьного типа или типа «дата», выводимое форматным модулем. Параметр может также быть опущен или иметь значение Nil.

Ширина – числовой параметр, ширина, занимаемая форматным модулем в строке формы, выраженная в процентах.

Шрифт – информация о размере, имени и аттрибутах шрифта данного модуля в виде символьной строки формата:
     "[Размер.Название шрифта] [Bold] [Italic]"
Аттрибуты – символьная строка, содержащая информацию о графических аттрибутах модуля. 

Возможны следующие значения:

L –линия по левой границе модуля
R –линия по правой границе модуля
T –линия по верхней границе модуля
B –линия по нижней границе модуля
F – обрамление модуля

Выравнивание – символьная строка, определяющая выравнивание данных внутри модуля. Может принимать одно из следующих значений:

L – выравнивание по левому краю модуля (значение по умолчанию)

R – выравнивание по правому краю модуля
C – выравнивание по центру модуля
Строка формы может состоять из нескольких форматных модулей. Если строка заканчивается символом “;”, то следующая строка считается продолженем данной. Форматные модули могут  располагаться только в секции [Begin]. Пустые строки в этой секции программного кода формы останутся пустыми и в результате выполнения формы.

Пример:


[Begin]


// Заголовок


{"Денежный поток", 50, "12. Arial Cyr Bold", "B", "C"}

    
{"Дата", 20, "10.Arial Cyr Bold", "F"};


{"Сумма", 30, "10.Arial Cyr Bold", "F"}
[End]

Управляющие структуры

DO WHILE
Синтаксис:

DO WHILE <условие>

  <программный код>...

END[DO]

Управляющая структура DO WHILE … ENDDO определяет многократно исполняемый блок программного кода, выполнение которого завершается в том случае, когда результатом выполнения указанного в предложении DO WHILE условия будет значение .F. (ложь). Такая структура называется циклом.

Пример:

[Declare]

Step:=1

[Begin]

DO WHILE Step <=5
  {"Step="+str(Step,1)}

  Step++
ENDDO


[End]



Результат выполнения:

Step=1


Step=2


Step=3


Step=4


Step=5


При помощи предложения DO WHILE можно организовывать вложенные циклы:

[Declare]

I

K

[Begin]

I:=1

DO WHILE I<=2
 K:=1

 DO WHILE K<=2
  {"I="+str(I,1), 10};   {"K="+str(K,1), 10}

  K++

 ENDDO

  I++
ENDDO


[End]


Результат выполнения:

I=1
K=1


I=1
K=2


I=2
K=1


I=2
K=2

INSERT DATA

Синтаксис:

INSERT DATA <имя БД> [FOR <select-функция>(<условие>)]

  <программный код>...

END

имя БД – имя базы данных внутри вызывающего программного модуля

select-функция – функция, производящая выборку из базы данных

Управляющая структура INSERT DATA … END определяет блок программного кода, который выполняется для каждой записи базы данных, удовлетворяющей указанному в select-функции предложения FOR условию. Условие должно представлять собой символьную строку, содержимое которой программа интерпритирует как исполняемый код. Если предложение FOR не указано, то исполнение программного кода происходит для всех записей базы данных.

Структуры INSERT DATA могут быть вложенными.

Пример:
Отчет о бухгалтерских операциях по кредиту счета 2620 

[Declare]

Account:="2620"

w1:=10

w2:=10

w3:=10

[Begin]

{" Date", w1+1, "Bold","LRBT"};

{" Document #", w1, "Bold","RBT"};

{" Amount", w1, "Bold","RBT"}

//

INSERT DATA Operat FOR;               SelectOperat("CrtAcc='"+Account+"'" )

  {, 1, ,"LB"};

  {Operat->Date, w1, ,"RB"};

  {" "+Operat->Code, w2, ,"RB"};

  {Operat->Sum, w3, ,"RB","R"}

END

[End]

Результат выполнения:

[image: image1.png]Date Document# _[Amount
06.05.2003 muTET 23000
06.05.2003 Tazraksts 0.15
07.05.2003 muTEl T35.70
06.05.2003 Tazraksts 045
06.05.2003 Taraksts 300
06.05.2003 muTEd 087
06.05.2003 muTE3 §0.47
09.05.2003 Tzraksts 0.75
09.05.2003 muT70 7340
09.05.2003 Tzraksts 771
09.05.2003 muTEg 95
09.05.2003 muTEs 5050
09.05.2003 muTE7 §1.29
09.05.2003 muTES T78.48
09.05.2003 muTEE 9979.92





INSERT TABLE
Синтаксис:

INSERT TABLE <имя БД> [FOR <select-функция>(<условие>)]

  <описание таблицы>
END

имя БД – имя базы данных внутри вызывающего программного модуля

select-функция – функция, производящая выборку из базы данных

Результат выполнения управляющей структуры INSERT TABLE … END представляет собой таблицу, каждая строка которой соответствует записи базы данных, удовлетворяющей указанному в select-функции предложения FOR условию. Условие должно представлять собой символьную строку, содержимое которой программа интерпритирует как исполняемый код. Если предложение FOR не указано, то исполнение программного кода происходит для всех записей базы данных.


Основное отличие структуры INSERT TABLE от описанной выше INSERT DATA состоит в том, что внутри нее содержится не непосредственно исполняемый код, а описание таблицы, на основании которого формируется ее графический образ и вставляется в готовую форму после выполнения всех исполняемых инструкций. 


Описание таблицы состоит из особых форматных модулей, каждый из которых занимает одну строчку исходного кода. Перенос строки посредством символа «;» в этом случае не допускается.


Каждый форматный модуль содержит описание одной колонки таблицы  и имеет следующий вид:

{<заголовок>,<данные>,<ширина>,[шрифт],[итог],[выравнивание]}
Заголовок – символьная строка, описывающая заголовок данной колонки. Если заголовок содержит нескольких элементов, располагаемых на разных строках, то эти элементы разделяются символом «|». Элементы заголовка выравниваются по левому краю колонки. Символ «~» перед элементом заголовка определяет выравнивание по центру.

При распечатке длинных таблиц заголовок автоматически переносится на каждый новый лист, содержащий таблицу.

Существует возможность указать общий заголовок для нескольких, описанных ниже колонок. В этом случае форматный модуль имеет следующую форму:

{<заголовок>, <количество колонок>}
Данные – символьная строка, содержащая исполняемый код. Как правило, это  ссылки на поля обрабатываемой базы данных. Внутри символьной строки могут использоваться ссылки на другие колонки данной записи таблицы, описанные перед данной. Это производится посредством обращения к встроенному массиву XREC: 

XREC[<номер столбца>]

Ширина – числовой параметр, ширина, занимаемая колонкой в таблице, выраженная в процентах.

Шрифт – информация о размере, имени и аттрибутах шрифта колонки в виде символьной строки формата:
     "[Размер.Название шрифта] [Bold] [Italic]"
Итог​ – символьная строка, исполняемая строка, логическая переменная или константа.  Определяет наличие строки итогов в нижней части таблицы. Если параметр или  результат вычисления исполняемой строки имеет  значение .T., то значение из данного элемента таблицы добавляется к итогу по колонке. Если параметр имеет символьный тип, то его значение размещается непосредственно в строке итогов.

Выравнивание – символьная строка, определяющая выравнивание данных внутри колонки. Может принимать одно из следующих значений:

L – выравнивание по левому краю модуля (значение по умолчанию)

R – выравнивание по правому краю модуля
C – выравнивание по центру модуля
Следует помнить, что данная структура может входить в программный код формы только один раз. 

Пример:

Отчет о бухгалтерских операциях по кредиту счета 2620 

[Declare]

Account:="2624"

w1:=10

w2:=10

w3:=10

[Begin]

INSERT TABLE Operat FOR;

 SelectOperat(" Operat->CrtAcc='"+Account+"'" )

  {"Date||~1",  "Operat->Date", w1,,"Total:"}

  {"Document",2}

  {"number|~2", "Operat->Code", w2}

  {"amount|~3", "Operat->Sum", w3,"Italic",.T.}

END
[End]

Результат:


Существует возможность влиять на наличие или отсутствие вертикальных и горизонтальных разделительных линий в результате исполнения INSERT TABLE. Для этого служат параметры секции [Control] VertDelimiter и HorzDelimiter соответственно.

VertDelimiter= Yes | No

HorzDelimiter= Yes | No

Пример:
[Control]

VertDelimiter=No

HorzDelimiter=Yes

В этом случае, в результате исполнения будут отсутствовать вертикальные разделители.

Список функций


Условная функция IF

Синтаксис:

If (<условие>, <выражение1>, <выражение2>) 


Функция вычисляет логическое выражение, указанное в качестве условия, если результат вычисления «истина», то выполняется выражение1, в противном случае выполняется выражение2.  

Пример:

Msg := If (x>=0, "Больше или равно 0", "Меньше 0") 

Функции обработки числовых величин


Abs (<параметр>)


Возвращает модуль, абсолютное значение параметра.


Int (<значение>)


Конвертирует значение в целое число путем отбрасывания дробной части.


Round (<значение>, <количество знаков>)

Округляет значение до указанного количества знаков. Если параметр количество знаков меньше нуля, то округляется целая часть числа.

Max (<значение1>, <значение2>)


Определяет максимальное из двух числовых значений или значений типа «дата».


Min (<значение1>, <значение2>)


Определяет минимальное из двух числовых значений или значений типа «дата».


Sqrt (<значение>)


Возвращает квадратный корень от указанного значения.


Chr (<параметр >)


Конвертирует числовое значение в символ ASCII. Параметр может принимать значения от 0 до 255.

Str (<числовое значение>, [общая длина строки], [длина дробной части])


Переводит числовое значение в строку указанной длины. Если общая длина строки  больше реальной длины полученного числа, то результат дополняется пробелами слева. 

DigitWrd (<параметр>)

Возвращает строку, содержащую значение параметра прописью. Параметр должен быт целой величиной.

Пример:

DigitWrd(120)

 Результат – “Viens simts divdesmit”

SW ([сумма], [длина строки], [денежная единица], [монета], [код языка])

Возвращает строку указанной длины, содержащую сумму прописью в определенной денежной единице. Умалчиваемое значение длины строки равно 100. Если опущен параметр сумма, то функция возвращает ту часть строки, которая не поместилась в длину строки, указанную в предыдущем вызове функции.

Код языка​ – язык, на котором выводится результат.

”LV” – Латышский, значение по умолчанию 

“RU” – Русский

“EN” – Английский

Денежная единица, монета – символьные строки, определяающие, в какой денежной единице указывается сумма. Умалчиваемое значение – латвийский лат. Для денежных единиц “USD” и “EUR” автоматически определяется монета – «цент.» или «cent.», в зависимости от выбранного языка.
Функции обработки символьных величин


Len (<символьная строка>)


Возвращает длину символьной строки.


Val (<символьная строка>)


Переводит число, записанное в символьной строке в числовое значение.


SubStr (<символьная строка>, <начальная позиция>, [количество символов])


Выделяет из символьной строки подстроку. Параметр начальная позиция определяет позицию подстроки внутри строки. Параметр количество символов определяет длину подстроки, если он опущен, то в подстороку будут входить все символы от начальной позиции до конца строки.


Left (<символьная строка>, <количество символов>)


Выделяет из символьной строки указанное количество символов слева.


Right (<символьная строка>, <количество символов>)


Выделяет из символьной строки указанное количество символов справа.


Space (<количество символов>)


Возвращает строку указанной длины, состоящую из пробелов.


Replicate (<символьная строка>, <количество повторений>)


Возвршает строку, состоящую из символьной строки, повторенной указанное количество раз.


StrTran (<символьная строка>, <заменяемая подстрока>, [замещающая подстрока])


Заменяет в  символьной строке все вхождения заменяемой подстроки на замещающую подстроку. Если замещающая подстрока  не указана, то заменяемая подстрока удалеется внутри  символьной строки.


AT (<символьная строка>, <подстрока>, [начальная позиция поиска])

Возвращает номер позиции первого символа подстроки в символьной строке.  


Пример:


AT("abcde", "cd")


Результат: 3


RAT (<символьная строка>, <подстрока>, [начальная позиция поиска])

Возвращает номер позиции последнего символа подстроки в символьной строке. 


TToken (<символьная строка>, <строка-разделитель>, <номер подстроки>)


Делит символьную строку на элементы, разделяемые строкой-разделителем и возвращает подстроку под указанным номером


Пример:


TToken("one; two; three", ";", 2)


Результат: “two”


StoA(<символьная строка>, <строка-разделитель>)


Преобразует символьную строку в массив на основании строки-разделителя.


Пример:


STOA("one; two; three", ";")


Результат: {"one”, “two”, “three"}


Upper (<символьная строка>) или LRUpper (<символьная строка>)

Преобразует символы, входящие в символьную строку из строчных в прописные.

LRUpper отличается тем, что позволяет обрабатывать символы русского и латышского языка.



Пример:


Upper("abcd")


Результат: "ABCD”


AllTrim (<символьная строка>)


Возвращает символьную строку с удаленными слева и справа пробелами.


LTrim (<символьная строка>)


Возвращает символьную строку с удаленными слева пробелами.

RTrim (<символьная строка>)


Возвращает символьную строку с удаленными справа пробелами.

PADС (<выражение>, <длина строки>)


PADL (<выражение>, <длина строки>)


PADR (<выражение>, <длина строки>)


Преобразует выражение символьного, цифрового типа или типа «дата» в символьную строку указанной длины, выравнивая данные по центру строки (PADC), по левому краю (PADL) или по правому краю (PADR).

Функции обработки времени и дат

Time () 
Возвращает строку, содержащую системное время.

Date () 
Возвращает системную дату.

Year (<дата>) 
Возвращает год указанной даты в виде четырехзначного числа.

Month (<дата>) 
Возвращает номер месяца указанной даты.

DTOC (<дата>)

Переводит дату в форматированную строку. Стандартный формат: “DD.MM.YYYY”

DTOS (<дата>)

Переводит дату в строку вида “YYYYMMDD”

CTOD (<форматированная строка>)

Переводит форматированную строку в дату. 
STOD (<строка>)

Переводит строку вида “YYYYMMDD” в дату.

EOP()

Возвращает текущую дату программного модуля, выполняющего форму.

PBD ()

Возвращает дату начала текущего периода программного модуля.

PTD ()

Возвращает последнюю дату текущего периода программного модуля.

CurrentYear ()

Возвращает текущий год программного модуля.

CurrentYearBegDate ()

Возвращает первую дату текущего года.

NextYearBegDate ()

Возвращает первую дату следующего года программного модуля.

Quarter (<дата>)

Возвращает номер квартала, в который входит указанная дата.

DOY (<дата>)

Возвращает порядковый номер указанного дня в году.

PeriodName([язык],[номер периода])

Возвращает название текущего периода программного модуля на языке системы или на указанном языке (1-русский, 2-латышский). Возможно также указание порядкового номера периода в списке периодов программы. 

DateWrd (<дата>)
Возвращает символьную строку, которая содержит дату, написанную словами.

MonthWrd (<дата>)

Возвращает символьную строку, состоящую из года и названия месяца, в который входит указанная дата.

AddMonth (<начальная дата>, <количество месяцев>)

Возвращает дату, отстоящую от начальной даты на указанное количество месяцев. 

